

LOGOPROGRAMM AUTORENNEN

```
PR Autos_löschen :x :y :Farben :Anzahl :Grösse
  Lokal "Laufindex"
  Setze "LaufIndex" 1
  Solange [:LaufIndex<=:Anzahl] [
    Auto_zeichnen Element :Laufindex :x Element :Laufindex :y [255 255 255] :Grösse
    Setze "Laufindex" :Laufindex + 1
  ]
ENDE

PR Autos_zeichnen :x :y :Farben :Anzahl :Grösse
  Lokal "Laufindex"
  Setze "LaufIndex" 1
  Solange [:LaufIndex<=:Anzahl] [
    Auto_zeichnen Element :Laufindex :x Element :Laufindex :y Element :Laufindex :Farben :Grösse
    Setze "Laufindex" :Laufindex + 1
  ]
ENDE

PR Auto_zeichnen :xpos :ypos :Farbe :Faktor

  ;* Farbe für das Auto mit Nummer "nr" setzen:
  Farbe Element 1 :Farbe Element 2 :Farbe Element 3 :Farbe

  ;* Setze Igel an Position (xpos,ypos): linke untere Ecke des Autos
  SH
  Aufxy :xpos :ypos
  Aufkurs 90
  SA
  ;* Zeichne Rumpf des Autos
  VW 30*:Faktor
  Links 90
  VW 5*:Faktor
  Links 45
  VW 5*:Faktor*QW(2)
  Links 45
  VW 25*:Faktor
  RW 20*:Faktor

  ;* Zeichne Cockpit des Autos
  Rechts 45
  VW 5*:Faktor*QW(2)
  Links 45
  VW 10*:Faktor
  Links 45
  VW 5*:Faktor*QW(2)
  Links 45

  ;* Zeichne Räder des Autos
  VW 10 *:Faktor
  Links 90
  VW 3*:Faktor
  Rechts 45
  VW 3*:Faktor*QW(2)
  Links 45
  VW 3*:Faktor
  Links 45
  VW 3*:Faktor*QW(2)
  Rechts 45
  VW 6*:Faktor
  Rechts 45
  VW 3*:Faktor*QW(2)
  Links 45
  VW 3*:Faktor
  Links 45
  VW 3*:Faktor*QW(2)
  Rechts 45
ENDE

PR Start_Ziel_zeichnen :Start :Ziel :Grösse
  Farbe 0 0 0
  SH
  Aufxy :Start+30*:Grösse (Element 1 :y)+20*:Grösse
  SA
  Aufxy :Start+30*:Grösse Element :Anzahl :y
  SH
  Aufxy :Ziel+30*:Grösse (Element 1 :y)+20*:Grösse
  SA
  Aufxy :Ziel+30*:Grösse Element :Anzahl :y
ENDE
```

```

PR START
;* Konstante für Zielkoordinate
Lokal "Ziel" "Start" "Grösse"
Setze "Grösse" 2

Setze "Start" 100
Setze "Ziel" :Start+200*:Grösse

Lokal "Anzahl"          ;* Anzahl der Autos
Lokal "Farbe"           ;* Zufallsfarbe für ein Auto
Lokal "x" "y" "Farben"  ;* Position und Farben für alle Autos (Listen)
Lokal "xHilf"           ;* temporäre Kopie der x-Koordinaten der Autos
Lokal "LaufIndex"       ;* Laufvariable
Lokal "Gewinner"        ;* Auto-nr des Gewinners

VersteckIgel
LöscheBild
StarteZufall

;* Anzahl der Autos eingeben lassen
Setze "Anzahl" ER Eingabefenster "Wieviel Spieler nehmen am Rennen teil?"

;* Listen für Position und Farben der Autos erstellen
Setze "Laufindex" 1
Solange [:LaufIndex<=:Anzahl] [
  Setze "y" ME 50*:LaufIndex :y
  Setze "x" ME :Start :x
  Setze "Farbe" []
  WH 3 [Setze "Farbe" ME ZZ 200 :Farbe]
  Setze "Farben" ML :Farbe :Farben
  Setze "LaufIndex" :LaufIndex+1
]

;* Startaufstellung zeichnen

Start_Ziel_zeichnen :Start :Ziel :Grösse
Autos_zeichnen :x :y :Farben :Anzahl :Grösse

;* Auf Start des Benutzers warten
Meldungsfenster "Bitte starten Sie das Rennen!" 1

;* Solange noch kein Auto das Ziel überquert hat werden die Autos weitersgesetzt.
Setze "Gewinner" 1

Solange [(Element :Gewinner :x)<=:Ziel] [
  Autos_löschen :x :y :Farben :Anzahl :Grösse          ;* und Autos neu zeichnen

  WH :Anzahl [Setze "x" ML ((ER :x) + (Zufallszahl 10)) (OE :x)]          ;* x-Koordinate aller
Autos wird erhöht um ZZ

  Autos_zeichnen :x :y :Farben :Anzahl :Grösse          ;* und Autos neu zeichnen

  Setze "LaufIndex" 1
  Solange [:Laufindex<=:Anzahl] [Wenn (Element :LaufIndex :x)>(Element :Gewinner :x) Dann (Setze
"Gewinner" :LaufIndex) Setze "LaufIndex" :LaufIndex+1]
]

Meldungsfenster (Wort "Der Spieler mit der Nummer " :Gewinner " hat gewonnen!!!") 1

ZeigIgel
ENDE

```